
Proposed change affects: UICC apps!  ME  Radio Access Network X Core Network X 
 

 
Title: ! Correct DN string representation formal definition to support wildcard character 
  
Source: ! SA5 (Nortel – Suzèle Lariven – lariven@nortel.com) 
  
Work item code: ! OAM-NIM  Date: ! 13/05/2005 
     
Category: ! F  Release: ! Rel-6 
 Use one of the following categories: 

F  (correction) 
A  (corresponds to a correction in an earlier release) 
B  (addition of feature),  
C  (functional modification of feature) 
D  (editorial modification) 

Detailed explanations of the above categories can 
be found in 3GPP TR 21.900. 

Use one of the following releases: 
Ph2 (GSM Phase 2) 
R96 (Release 1996) 
R97 (Release 1997) 
R98 (Release 1998) 
R99 (Release 1999) 
Rel-4 (Release 4) 
Rel-5 (Release 5) 
Rel-6 (Release 6) 

     Rel-7 (Release 7) 
 

  
Reason for change: ! Correct DN string representation formal definition to support wildcard character 
  
Summary of change: ! • Precision that the wildcard character is to represent one or more characters. 

• Modification & extension of existing EBNF syntax-rules and addition of new 
EBNF syntax-rules in order to allow presence of wildcard characters in place 
of any sequence of one or more characters in any of the following: 

• DN prefix RFC 1035 attribute value domain component labels 
• DN prefix RFC 2253 attribute types and attribute values 
• IRPAgent local DN attribute type class and naming attribute names 
• IRPAgent local DN attribute values 

• Modification of EBNF syntax-rule in order not to allow presence of separator 
character for class and naming attribute of IRPAgent local DN attribute type in 
names of these classes and naming attributes 

• Editorial corrections 
  
Consequences if  ! 
not approved: 

The TS would be inconsistent as regards support of wildcard character. 

  
Clauses affected: ! 7.2, 7.3 
  
 Y N   
Other specs !  X  Other core specifications !  
affected:  X  Test specifications  
  X  O&M Specifications  
  
Other comments: !  

Change in Clause 7.2 

7.2 Character syntax 
[…] 

3. The asterisk character ('*', U+002A) is reserved to denote wildcard. Wildcard character(s) can appear in 
AttributeType and AttributeValue. The wildcard character can be used to represent one or more 
characters. 

End of Change in Clause 7.2 
 

Change in Clause 7.3 

7.3 EBNF of DN String Representation 
The formal definitions provided within this subclause consolidate several rules and concepts (null distinguished name, 
DN prefix, local DN, domain component type, class names starting with upper case characters, attribute names starting 
with lower case characters, classes with or without an "Id" naming attribute, attribute type and attribute value allowed 
characters, wildcard character). The definition is more detailed to clarify these naming rules, and will not introduce 
compliancy issues for implementations compliant with Rel-5 version of this specification. 

The following is the EBNF for DN in string representation (Extended Backus-Naur Form; see ISO/IEC 14977 [13] for 
more information): 

DistinguishedName              =   NullDN       (* Distinguished Names shall not exceed   *) 
                                 | RegularDN ;  (* 400 octets as specified in section 7.4 *) 
 
NullDN                         = ;  (* empty string; null DN is specified in subclause 7.1.1 *) 
 
RegularDN                      =   DNPrefixPlusRDNSeparator  (* DN prefix and local DN *) 
                                 , LocalDN ;                 (* are defined in annex C *) 
 
DNPrefixPlusRDNSeparator       =   ( NullDNPrefix , NullRDNSeparator ) 
                                 | ( DNPrefixWithDomainComponent , RDNSeparator ) 
                                 | ( DNPrefixWithoutDomainComponent , RDNSeparator ) ; 
 
NullDNPrefix                   = ;  (* empty string *) 
 
NullRDNSeparator               = ;  (* empty string *) 
 
DNPrefixWithDomainComponent    =   DomainComponentRDN 
                                 , { RDNSeparator , DomainComponentRDN } 
                                 , { RDNSeparator , RegularRDN } ; 
 
DNPrefixWithoutDomainComponent =   RegularRDN 
                                 , { RDNSeparator , RegularRDN } ; 
 
LocalDN                        =   LocalRDN 
                                 , { RDNSeparator , LocalRDN } ; 
 
RDNSeparator                   =   [ RDNSeparatorWhiteSpace ]    (* use of optional white space  *) 
                                 , CommaChar                     (* is recommended to be avoided *) 
                                 , [ RDNSeparatorWhiteSpace ] ; 
 
RDNSeparatorWhiteSpace         =   [ CarriageReturnChar ] 
                                 , { SpaceChar } ; 
 
DomainComponentRDN             = DCAttributeTypeAndValue ; 
 
RegularRDN                     = RegularAttributeTypeAndValue ; 
 
LocalRDN                       = LocalDNAttributeTypeAndValue ; 
 
DCAttributeTypeAndValue        =   DCAttributeType 
                                 , AttributeTypeAndValueSeparator 
                                 , ( DCAttributeValue | WildcardDCAttributeValue ) ; 

RegularAttributeTypeAndValue   =   ( RegularAttributeType | WildcardRegularAttributeType ) 
                                 , AttributeTypeAndValueSeparator 
                                 , ( RegularAttributeValue | WildcardRegularAttributeValue ) ; 
 
LocalDNAttributeTypeAndValue   =   ( LocalDNAttributeType | WildcardLocalDNAttributeType ) 
                                 , AttributeTypeAndValueSeparator 
                                 , ( RegularAttributeValue | WildcardRegularAttributeValue ) ; 
 
AttributeTypeAndValueSeparator = EqualsSignChar ; 
 
DCAttributeType                = "DC" ;  (* ISO/IEC 646 IRV U+0044/0043 Latin capital letters D&C *) 
 
DCAttributeValue               =   DCLabel                           (* this is specified *) 
                                 , { FullStopChar , DCLabel } ;  (* in IETF RFC 1035  *) 
                                 , { DCLabelSeparator , DCLabel } ;  (* in IETF RFC 1035  *) 
 
WildcardDCAttributeValue       =   (   ( DCLabel | WildcardDCLabel ) 
                                     , { DCLabelSeparator , ( DCLabel | WildcardDCLabel ) } ) 
                                 - DCAttributeValue ; 
 
DCLabelSeparator               = FullStopChar ;  (* this is specified in IETF RFC 1035 *) 
 
DCLabel                        =   LetterChar                             (* this is specified *) 
                                 , [   { LetterDigitHypenMinusChar }      (* in IETF RFC 1035  *) 
                                     , LetterDigitChar               ] ; 
 
WildcardDCLabel                =   (   ( LetterChar | WildcardChar ) 
                                     , [   { LetterDigitHypenMinusChar | WildcardChar } 
                                         , ( LetterDigitChar | WildcardChar )           ] ) 
                                 - DCLabel ; 
 
RegularAttributeType           =   LetterChar                       (* this is specified *) 
                                 , { LetterDigitHypenMinusChar } ;  (* in IETF RFC 2253  *) 
 
WildcardRegularAttributeType   =   (   ( LetterChar | WildcardChar ) 
                                     , { LetterDigitHypenMinusChar | WildcardChar } ) 
                                 - RegularAttributeType ; 
 
LocalDNAttributeType           =   NameOfClassWithIdAttribute        (* definition selected shall *) 
                                 | NamesOfClassAndNamingAttribute ;  (* be in accordance with the *) 
                                                                     (* rules defined in annex B  *) 
 
WildcardLocalDNAttributeType   =   WildcardNameOfClassWithIdAttr 
                                 | WildcardNamesOfClassAndNamAttr ; 
 
NameOfClassWithIdAttribute     = ClassName ;  (* see rules defined in annex B *) 
 
WildcardNameOfClassWithIdAttr  = WildcardClassName ; 
 
NamesOfClassAndNamingAttribute =   ClassName                      (* see rules defined in annex B *) 
                                 , FullStopChar 
                                 , ClassNamingAttributeSeparator 
                                 , NamingAttributeName ; 
 
WildcardNamesOfClassAndNamAttr =   (   ( ClassName | WildcardClassName ) 
                                     , ClassNamingAttributeSeparator 
                                     , ( NamingAttributeName | WildcardNamingAttributeName ) ) 
                                 - NamesOfClassAndNamingAttribute ; 
 
ClassNamingAttributeSeparator  = FullStopChar ;  (* see rules defined in annex B *) 
 
ClassName                      =   CapitalLetterChar               (* see recommendation on       *) 
                                 , { LocalDNAttributeTypeChar } ;  (* characters for class names  *) 
                                                                   (* in 3GPP TS 32.622 annex A   *) 
 
WildcardClassName              =   (   ( CapitalLetterChar | WildcardChar ) 
                                     , { LocalDNAttributeTypeChar | WildcardChar } ) 
                                 - ClassName ; 
 
NamingAttributeName            =   SmallLetterChar 
                                 , { LocalDNAttributeTypeChar } ; 
 
WildcardNamingAttributeName    =   (   ( SmallLetterChar | WildcardChar ) 
                                     , { LocalDNAttributeTypeChar | WildcardChar } ) 
                                 - NamingAttributeName ; 

RegularAttributeValue          =   ( AttributeValueChar – SpaceChar )          (* this is       *) 
                                 , [   { AttributeValueChar }                  (* specified in  *) 
                                     , ( AttributeValueChar – SpaceChar ) ] ;  (* IETF RFC 2253 *) 
 
WildcardRegularAttributeValue  =   (   ( ( AttributeValueChar – SpaceChar ) | WildcardChar ) 
                                     , [   { AttributeValueChar | WildcardChar } 
                                         , ( ( AttributeValueChar – SpaceChar ) | WildcardChar ) ] ) 
                                 - RegularAttributeValue ; 
 
LocalDNAttributeTypeChar       = DNChar – FullStopChar ; 
 
AttributeValueChar             = DNChar | EscapedCharSequence ; 
 
WildcardChar                   = AsteriskChar ;  (* this is specified in subclause 7.2 *) 
 
DNChar                         = DNCharUnrestricted - ReservedChar ; 
 
DNCharUnrestricted             =     ? Character of ISO/IEC 646 IRV ? 
                                   | ? Character of standard coded character set 
                                       supporting and extending ISO/IEC 646 IRV  ? ; 
 
EscapedCharSequence            =   ReverseSolidusChar          (* this is specified  *) 
                                 , 2 * HexadecimalDigitChar ;  (* in subclause 7.1.3 *) 
 
ReservedChar                   =   Rfc2253ReservedChar | CarriageReturnChar | LineFeedChar 
                                 | AsteriskChar ; 
 
Rfc2253ReservedChar            =   CommaChar | EqualsSignChar | PlusSignChar | LessThanSignChar 
                                 | GreaterThanSignChar | NumberSignChar | SemiColonChar 
                                 | ReverseSolidusChar | QuotationMarkChar ; 
 
LetterChar                     = CapitalLetterChar | SmallLetterChar ; 
 
LetterDigitChar                = LetterChar | DigitChar ; 
 
LetterDigitHypenMinusChar      = LetterDigitChar | HypenMinusChar ; 
 
HexadecimalDigitChar           = DigitChar | CapitalLetterAtoFChar | SmallLetterAtoFChar ; 
 
LineFeedChar                   = ? ISO/IEC 646 IRV U+000A character line feed ? ; 
 
CarriageReturnChar             = ? ISO/IEC 646 IRV U+000D character carriage return ? ; 
 
SpaceChar                      = ' ' ;  (* ISO/IEC 646 IRV U+0020 character space *) 
 
QuotationMarkChar              = '"' ;  (* ISO/IEC 646 IRV U+0022 character quotation mark *) 
 
NumberSignChar                 = '#' ;  (* ISO/IEC 646 IRV U+0023 character number sign *) 
 
AsteriskChar                   = '*' ;  (* ISO/IEC 646 IRV U+002A character asterisk *) 
 
PlusSignChar                   = '+' ;  (* ISO/IEC 646 IRV U+002B character plus sign *) 
 
CommaChar                      = ',' ;  (* ISO/IEC 646 IRV U+002C character comma *) 
 
HypenMinusChar                 = '-' ;  (* ISO/IEC 646 IRV U+002D character hyphen-minus *) 
 
FullStopChar                   = '.' ;  (* ISO/IEC 646 IRV U+002E character full stop *) 
 
DigitChar                      =   '0' | '1' | '2' | '3' | '4'    (* ISO/IEC 646 IRV U+0030-0039 *) 
                                 | '5' | '6' | '7' | '8' | '9' ;  (* digits 0 to 9               *) 
 
SemiColonChar                  = ';' ;  (* ISO/IEC 646 IRV U+003B character semicolon *) 
 
LessThanSignChar               = '<' ;  (* ISO/IEC 646 IRV U+003C character less-than sign *) 
 
EqualsSignChar                 = '=' ;  (* ISO/IEC 646 IRV U+003D character equals sign *) 
 
GreaterThanSignChar            = '>' ;  (* ISO/IEC 646 IRV U+003E character greater-than sign *) 
 
CapitalLetterAtoFChar          =   'A' | 'B' | 'C'    (* ISO/IEC 646 IRV U+0041-0046  *) 
                                 | 'D' | 'E' | 'F' ;  (* Latin capital letters A to F *) 
 
CapitalLetterChar              =   CapitalLetterAtoFChar | 'G' | 'H'    (* ISO/IEC 646 IRV *) 
                                 | 'I' | 'J' | 'K' | 'L' | 'M' | 'N'    (* U+0041-005A     *) 
                                 | 'O' | 'P' | 'Q' | 'R' | 'S' | 'T'    (* Latin capital   *) 
                                 | 'U' | 'V' | 'W' | 'X' | 'Y' | 'Z' ;  (* letters A to Z  *) 

ReverseSolidusChar             = '\' ;  (* ISO/IEC 646 IRV U+005C character reverse solidus *) 
 
SmallLetterAtoFChar            =   'a' | 'b' | 'c'    (* ISO/IEC 646 IRV U+0061-0066 *) 
                                 | 'd' | 'e' | 'f' ;  (* Latin small letters a to f  *) 
 
SmallLetterChar                =   SmallLetterAtoFChar | 'g' | 'h'      (* ISO/IEC 646 IRV *) 
                                 | 'i' | 'j' | 'k' | 'l' | 'm' | 'n'    (* U+0061-007A     *) 
                                 | 'o' | 'p' | 'q' | 'r' | 's' | 't'    (* Latin small     *) 
                                 | 'u' | 'v' | 'w' | 'x' | 'y' | 'z' ;  (* letters a to z  *) 

End of Change in Clause 7.3 
End of Document 

Annex E (informative): 
Change history 

Change history 
Date TSG # TSG Doc. CR Rev Subject/Comment Old New 

Jun 2001 SA_12 SP-010283 -- -- Approved at TSG SA #12 and placed under Change Control 2.0.0 4.0.0 
Dec 2001 SA_14 SP-010641 001 -- Alignment of Figure C.1 with text in annex C 4.0.0 4.1.0 
Sep 2001 SA_17 SP-020481 002 -- Upgrade to Rel-5 (Remove information in the Introduction that is 

only relevant to Rel-4) 
4.1.0 5.0.0 

Dec 2002 -- -- -- -- Cosmetics 5.0.0 5.0.1 
Dec 2004 SA_26 SP-040793 003 -- Correct and convert formal specification from BNF syntax to EBNF 

with corrections 
5.0.1 6.0.0 

Mar 2005 SA_27 SP-050049 004 -- Re-introduce text erroneously deleted during implementation of CR 
003 

6.0.0 6.1.0 

